

Search Query Grammar

Intro

This document provides a BNF description for YouTrack Search Query grammar.

Grammar

```
<SearchRequest> ::= <Term> ('or' <Term>)*
<Term> ::= <TermItem>*
<TermItem> ::= <QuotedText> | <NegativeText> | <PositiveSingleValue> |
<NegativeSingleValue> | <Sort> | <CategorizedFilter> | <Text>
<CategorizedFilter> ::= <Attribute> ':' <AttributeFilter> (',' <AttributeFilter>)*
<AttributeFilter> ::= ('-'? <Value> ) | ('-'? <ValueRange>)
<ValueRange> ::= <Value> '..' <Value>
<PositiveSingleValue> ::= '#'<SingleValue>
<NegativeSingleValue> ::= '-'<SingleValue>
<SingleValue> ::= <Value>
<Sort> ::= 'sort by:' <SortField> (',' <SortField>)*
<SortField> ::= <SortAttribute> ('asc' | 'desc')?
<QuotedText> ::= '"' <text without quotes> '"'
<NegativeText> ::= '-' <QuotedText>
<Text> ::= <any text>
<Value> ::= <ComplexValue> | <SimpleValue>
<SimpleValue> ::= <value without spaces>
<ComplexValue> ::= '{' <value (can have spaces)> '}'
```

Grammar is case-insensitive.

For complete list of search attributes please refer the [Search and Command Attributes](#).

Samples

As new reach text area component is coming we can use it for highlighting of queries instead of weird underlinings.

Query	Description
"Who is mister Putin"	Issues that contain phrase "Who is mister Putin"
for: me #unresolved	Unresolved issues assigned to the current user
for: me unresolved	Issues assigned to the current user and contain word "unresolved"
for: mazine	Issues assigned to user with login "mazine"
for: batman	Issues assigned to user with login "batman" when there is no user "batman"
mazine	Issues that contain word "mazine"
#mazine	Issues created by or assigned to user with login "mazine"
-mazine	Issues created by or assigned to any user but "mazine"
state: -open, -submitted	Issues in any state but "open" or "submitted"
sort by: priority, issue id asc	All issues sorted by priority in default order and then by issue id in ascending order